

## 現場のための水文学 (5)

### - 流出解析 その5 -

若手水文学研究会\*

#### 5. 非線形貯留方程式による流出解析

降雨と流出の概念については第2章で述べたが、これまでは降雨に線形に応答した形で流出が生起するようなモデルで流出現象を記述してきた。しかしながら、現実には流出の過程では浸透や蒸発散など複雑な物理的作用が働き、常に降雨と流出の関係が線形系で表されるとは言いがたい。特にそのような物理的特性が大きく反映される小流域の流出パターンについては、非線形性が強く現れる可能性もある。そこで、本報告ではこれまで培った種々のテクニックを駆使し、非線形貯留方程式を用いた流出解析を考えていきたい。

例として一価と二価の非線形貯留方程式を用いた流出解析手法を示すが、結局はこれらの非線形方程式を線形化し、先に示した線形貯留方程式で扱ったものと同様な方法論に帰着させて流出解析を行なっている。なお、解説の中では必要な分量の数式が掲載されている。また、後にはマトリックス計算もあり、これは苦手としている方も多かろう。そのような方は5-1節がスカラー計算で比較的理解しやすいかと思うので、そこで基本的な考え方を掴んでほしい。後段についてはマトリックスもスカラーと同様な考え方で取り扱えるという理解だけで結構である。もしくは、ポイントとなる式にはアンダーラインを引いておいたのでそれに注目してもらいたい。実際の計算手順を理解するために、補遺にはPC-98クラスのパソコンで対処しうるようにBASICで組んだプログラムを掲載してある。さらに、式があるだけで嫌悪感をもつ方にも、結果だけは出せるようなメニューも用意してある。

#### 5-1 一価の非線形貯留方程式による流出解析

##### 5-1-1 線形近似による解法

最初に最も簡単な例として一価の非線形貯留方程式を考える。その基礎式を次に示す。

$$s = kq^p \quad (5.1)$$

$$\frac{ds}{dt} = r - q \quad (5.2)$$

ここで、 $s$ は貯留量(mm)、 $r$ は有効雨量(mm/h)、 $q$ は流出高(mm/h)、 $k$ および $p$ はモデルパラメータである。これまでの報告では貯留量と流出高の関係( $s-q$ 関係)に線形関係をあてはめて流出計算を行ってきた。ところが、図5.1に示すような実際の $s-q$ 関係をみると、必ずしも線形関係(両対数関係の勾配が1)を当てはめることが妥当とは言いきれない。なお

\*星 清(水工部長), 鳥谷部寿人, 金高州吾, 三浦敦禎, 市川嘉輝(河川研究室), 中津川 誠, 谷 昭彦, 山口昌志(環境研究室), 佐伯礼子(北開水工コンサルタント), 白川俊也, 柴田春幸(松木設計事務所)

本例は、1992年8月の鶴川の洪水事例からつくられたもので、詳細は第2章を参照されたい。図中に点線で示したのが線形関係を当てはめたもので、式(5.1)中の $p$ を1としたものに相当する。また、実線は $p$ を0.7としたもので、こちらの方が全体的な $s-q$ 関係に沿っているように見える。ただし、ここでは $s-q$ 関係に一価関数を当てはめており、二価性(ループを描くこと)を再現することはできない。二価の非線形貯留方程式を用いた流出モデルについては後述する。




図 5.1 直接流出高と貯留量の関係  
(鶴川, 1992年8月9日の事例)

さて、実際に上式を解くわけだが、数式処理の便宜を考え、次のような変数変換をおこなう。

$$y = q^p \quad (5.3)$$

そのうえで式(5.2)に式(5.1)を代入すると、

$$\frac{dy}{dt} = -\frac{1}{k}y^{1/p} + \frac{1}{k}r \quad (5.4)$$

となる。これは非線形の常微分方程式なので、ルンゲクッタ法など種々の既存数値解法が適用できる。しかしながら、本報では先に示した線形貯留方程式の解法がそのまま利用でき、しかも十分な計算精度が期待できる線形近似法を用いることとする。

一般にある関数 $f(y)$ が与えられたとき、これはテイラー級数展開で1次の項までとると、次のように近似できる。

$$f(y) = f(y^*) + a(y^*)(y - y^*) \quad (5.5)$$

ここで、 $y^*$ は1ステップ前の値をあらわす。また、 $a(y) = df(y)/dy$ である。いま、式(5.4)の右辺を $f(y)$ として式(5.5)のように線形化すると、

$$\frac{dy}{dt} = -\frac{1}{k} \frac{1}{p} (y^*)^{1/p-1} y - \frac{1}{k} \left(1 - \frac{1}{p}\right) (y^*)^{1/p} + \frac{1}{k} r \quad (5.6)$$

となる。これを整理すると、

$$\frac{dy}{dt} = ay + x \quad (5.7)$$

となり、これまで何度も取り扱ってきた1階線形常微分方程式に帰着させることができる。とにかくここまでたどり着けば、あとはこれまで示してきた解法が適用できる。詳細は前報を参照するとして、結果のみ示す。式(5.7)の数値解は次に示される。

$$y_{m+1} = \phi y_m + \gamma x_m \quad (5.8)$$

ここで、 $m$  は時間ステップ数である。また、各変数の内訳を表 5.1 に示す。

表 5.1 貯留方程式の数値解に関わる諸変数 (一価非線形)

$$\begin{aligned}
 a &= -\frac{1}{k} \frac{1}{p} y_m^{1/p-1} \\
 x_m &= -\frac{1}{k} \left(1 - \frac{1}{p}\right) y_m^{1/p} + \frac{1}{k} r_{m+1}, & d_m &= \frac{1}{k^2} [y_m^{1/p} - r_{m+1}] \\
 \phi &= 1 + aT + \frac{1}{2}(aT)^2 + \frac{1}{6}(aT)^3 + \frac{1}{24}(aT)^4, & \gamma &= T \left[1 + \frac{1}{2}(aT) + \frac{1}{6}(aT)^2 + \frac{1}{24}(aT)^3\right]
 \end{aligned}$$

なお、これを解いて得られた  $y$  から実際の流出高  $q$  を算定するには、 $q = y^{1/p}$  の変換をすればよいことは式 (5.3) から明らかである。ところで、ここで気づくのは線形方程式の場合、係数  $a$  および  $x$  は定数 (もしくは強制入力項) として与えられたのが、非線形方程式を線形化した場合には変数となって 1 ステップ前の値の影響を常に受けることである。したがって、離散化する場合の時間ステップは、線形近似による誤差の増大を招かぬよう十分小さくとる必要がある。非線形方程式を解く際に計算時間がかかるのはこのためである。

#### 5-1-2 ルンゲクッタギル法と線形近似法による結果の比較

前節で提示した非線形流出モデルの線形近似による解法が妥当な結果を与えるかをチェックしてみる。一般的に非線形常微分方程式を数値的に解くのに最も精度の高い解を与える方法としてルンゲクッタギル法がある。詳細については様々なテキスト<sup>4)</sup>で紹介されている。端的に言えば、線形近似法 (もしくはオイラー法) が式 (5.5) のように微分方程式を 1 次の項までテイラー級数展開したものであり、これに対し、2 次の項まで展開し、精度の向上を図ったのがルンゲクッタ法である。また、この際の計算ルーチンに工夫を加えたものがルンゲクッタギル法である。いま、次のような微分方程式が与えられた場合、ルンゲクッタギル法による解は以下に示される。

$$\frac{dy}{dt} = f(y) \quad (5.9)$$

$$y_{m+1} = y_m + \frac{1}{6} \left[ k_1 + (2 - \sqrt{2})k_2 + (2 + \sqrt{2})k_3 + k_4 \right] \quad (5.10)$$

ここで、 $k_1 = hf(y_m)$ ,  $k_2 = hf(y_m + \frac{1}{2}k_1)$ ,  $k_3 = hf(y_m + \frac{\sqrt{2}-1}{2}k_1 + \frac{2-\sqrt{2}}{2}k_2)$ ,  $k_4 = hf(y_m - \frac{\sqrt{2}}{2}k_2 + \frac{2+\sqrt{2}}{2}k_3)$  である。また、 $h$  は刻み幅である。ここで示された解は先に示した式 (5.8) と比較されたい。なお、具体的な計算手順については例題を通して理解していただきたい。

#### (例題 12)

式 (5.1)~(5.2) のような一価の非線形貯留方程式を用い、線形近似法とルンゲクッタギル法により流出高を算出して結果を比較せよ。ここで有効雨量は図 5.2 に示す鶴川の洪水事例 (数値は表 1.2 参照) を与え、パラメータ  $k$  は 20、 $p$  は 0.7 とせよ。

(例題 12 の解答)

ここでは時間刻みを 0.2hr とした。最初にルンゲクッタギル法による計算事例を示す。

例として  $m = 1$  の計算過程を示す。初期条件は  $q_0 = 0$  なので、 $y_0 = q_0^p = 0$  である。時間刻みが 0.2hr なので、1hr の結果を得るには  $t = 0$  からルーチンを 5 回繰り返す必要があり、最初に繰り返し 1 回目 (0.2hr) の計算結果を示す。

$$\begin{aligned}k_1 &= hf(y_0) = hf(0) = h \left( -\frac{1}{k} y_0^{1/p} + \frac{1}{k} r_1 \right) = 0.2 \left( 0 + \frac{1}{20} \cdot 0.776 \right) = 0.00776 \\k_2 &= hf(y_0 + k_1/2) = hf(0.00388) = 0.00776 \\k_3 &= hf(y_0 + (\sqrt{2} - 1)/2 \cdot k_1 + (2 - \sqrt{2})/2 \cdot k_2) = hf(0.00388) = 0.00776 \\k_4 &= hf(y_0 - \sqrt{2}/2 \cdot k_2 + (2 + \sqrt{2})/2 \cdot k_3) = hf(0.00776) = 0.00775 \\y_{0.2} &= y_0 + \frac{1}{6} [k_1 + (2 - \sqrt{2})k_2 + (2 + \sqrt{2})k_3 + k_4] = 0.00776\end{aligned}$$

続いて以下、0.4hr, 0.6hr, 0.8hr, 1hr の結果を示す。

$$\begin{aligned}t = 0.4hr : k_1 &= 0.00775, k_2 = 0.00774, k_3 = 0.00774, k_4 = 0.00773, y_{0.4} = 0.0155 \\t = 0.6hr : k_1 &= 0.00773, k_2 = 0.00772, k_3 = 0.00772, k_4 = 0.00771, y_{0.6} = 0.0232 \\t = 0.8hr : k_1 &= 0.00771, k_2 = 0.00770, k_3 = 0.00770, k_4 = 0.00769, y_{0.8} = 0.0309 \\t = 1.0hr : k_1 &= 0.00769, k_2 = 0.00768, k_3 = 0.00768, k_4 = 0.00766, y_{1.0} = 0.0386\end{aligned}$$

これによって 1hr における流出高は  $q_1 = y_1^{1/p} = (0.0386)^{1/0.7} = 0.01\text{mm/h}$  となる。次に線形近似法による計算過程を上記と同様に示す。

$$\begin{aligned}a &= -\frac{1}{k} \frac{1}{p} y_0^{1/p-1} = 0 \\ \phi &= 1 + aT + 1/2(aT)^2 + 1/6(aT)^3 + 1/24(aT)^4 = 1 \\ \gamma &= T[1 + 1/2(aT) + 1/6(aT)^2 + 1/24(aT)^3] = 0.2 \\ x_0 &= -\frac{1}{k} \left(1 - \frac{1}{p}\right) y_0^{1/p} + \frac{1}{k} r_1 = 0 + \frac{1}{20} \cdot 0.776 = 0.0388 \\ y_{0.2} &= \phi y_0 + \gamma x_0 = 0 + 0.2 \cdot 0.0388 = 0.00776\end{aligned}$$

続いて以下、0.4hr, 0.6hr, 0.8hr, 1hr の結果を示す。

$$\begin{aligned}t = 0.4hr : a &= -0.0089, \phi = 0.998, \gamma = 0.200, x_{0.2} = 0.0388, y_{0.4} = 0.0155 \\t = 0.6hr : a &= -0.0120, \phi = 0.998, \gamma = 0.200, x_{0.4} = 0.0389, y_{0.6} = 0.0232 \\t = 0.8hr : a &= -0.0142, \phi = 0.997, \gamma = 0.200, x_{0.6} = 0.0389, y_{0.8} = 0.0309 \\t = 1.0hr : a &= -0.0161, \phi = 0.997, \gamma = 0.200, x_{0.8} = 0.0389, y_{1.0} = 0.0386\end{aligned}$$

これによって 1hr における流出高は  $q_1 = y_1^{1/p} = (0.0386)^{1/0.7} = 0.01\text{mm/h}$  となり、先のルンゲクッタギル法で算出した結果と一致する。なお、その後の時間について両手法によって算出された結果を表 5.2 および図 5.2 に示してある。これらから、非線形常微分方程式の解法として線形近似法が十分な精度を保証していることがわかる。

表 5.2 ルンゲクッタギル法と線形近似法の結果の比較

時間 (hr)	有効雨量 (mm/h)	流出高 R. K. G. (mm/h)	流出高 線形近似 (mm/h)
0	0.000	0.000	0.000
1	0.776	0.010	0.010
2	1.878	0.055	0.055
3	1.640	0.107	0.107
4	0.626	0.126	0.126
5	0.119	0.126	0.126
6	1.909	0.198	0.198
7	9.251	0.679	0.679
8	5.783	1.014	1.014
9	3.430	1.185	1.185
10	10.064	1.875	1.876
11	18.589	3.392	3.392
12	14.421	4.519	4.520
13	10.909	5.213	5.213
14	20.467	6.975	6.976
15	7.836	7.079	7.079
16	2.773	6.565	6.565
17	0.507	5.861	5.862
18	0.000	5.202	5.203
19	0.000	4.637	4.637
20	0.031	4.151	4.152
21	0.000	3.728	3.728
22	0.000	3.358	3.358
23	0.000	3.035	3.035
24	0.000	2.751	2.751
25	0.000	2.501	2.501

時間 (hr)	有効雨量 (mm/h)	流出高 R. K. G. (mm/h)	流出高 線形近似 (mm/h)
26	0.000	2.279	2.280
27	0.000	2.083	2.083
28	0.000	1.908	1.908
29	0.000	1.751	1.751
30	0.000	1.611	1.611
31	0.000	1.485	1.485
32	0.000	1.372	1.372
33	0.000	1.269	1.269
34	0.000	1.176	1.176
35	0.000	1.092	1.092
36	0.000	1.016	1.016
37	0.000	0.946	0.946
38	0.000	0.883	0.883
39	0.000	0.825	0.825
40	0.000	0.771	0.771
41	0.000	0.722	0.723
42	0.000	0.678	0.678
43	0.000	0.636	0.636
44	0.000	0.598	0.598
45	0.000	0.563	0.563
46	0.000	0.530	0.530
47	0.000	0.500	0.500
48	0.000	0.472	0.472
49	0.000	0.446	0.446
50	0.000	0.422	0.422




図 5.2 ルンゲクッタギル法と線形近似法の結果の比較

### 5-1-3 最適化手法によるパラメータの同定

前節で示した一価の非線形貯留方程式のパラメータを最適化することを考える。この場合、先に示したようにパラメータは  $k$  と  $p$  であるが、指数部のパラメータについては物理的な推察から固定して考えることが望ましい。これは後で示す二価の貯留方程式についても

同様である。そこで、以下では  $k$  のみの 1 パラメータ問題として最適化をおこなう。最適化の考え方については既に前報に詳しく掲載しており、非線形貯留方程式にしても先に示したように線形化すれば全く同じようにそれが利用できる。

実測値との誤差を最小とするようにパラメータを逐次更新して最適値を得るのが最適化手法であるが、その更新量  $\Delta k$  は次のように得られる。

$$\Delta k = \frac{\sum e_i v_i}{\sum v_i^2} \quad (5.11)$$

ここで、 $N$  個あるデータのうち  $i$  番目のデータについて考えると、 $e_i$  は計算流出高  $q_i$  と実測流出高  $q_i^*$  の差で、 $e_i = q_i^* - q_i$  である。また、 $v_i$  は感度係数といわれ、計算流出高のパラメータに対する感度をあらわすもので、 $v_i = \partial q_i / \partial k$  で定義される。この感度係数を変数変換により、 $u_i = \partial y_i / \partial k$  とあらわす。この際の  $u$  と  $v$  の関係は次の通りである。

$$v_i = \frac{1}{p} y_i^{1/p-1} u_i \quad (5.12)$$

ところで、 $u_i$  であらわされる感度係数は、先に示したような貯留方程式と同様に 1 階線形の常微分方程式であらわされることは前報で示した。

$$\frac{du}{dt} = au + d \quad (5.13)$$

また、その数値解は、

$$u_{m+1} = \phi u_m + \gamma d_m \quad (5.14)$$

となる。ここで、 $d_m$  は表 5.1 に示してある。また、 $a$ 、 $\phi$  および  $\gamma$  は先に示したもの (表 5.1 参照) と全く同様である。すなわち、式 (5.7) と式 (5.13)、式 (5.8) と式 (5.14) は対をなすもので、ほとんど同じルーチンで流出高と感度係数が得られることがわかる。

結局、式 (5.8) により流出高  $q$ 、式 (5.14) により感度係数  $u$  を求め、式 (5.11) によりパラメータ  $k$  を更新し、最適値が得られるまで計算を繰り返すのが最適化によるパラメータ同定法である。繰り返し計算の打ち切り条件は次の通りである。

$$\left| \frac{\Delta k_n}{k_n} \right| < \varepsilon \quad (5.15)$$

ここで、 $\varepsilon$  は収束許容限界値で通常  $10^{-2} \sim 10^{-4}$  に設定する。また、 $n$  は繰り返しステップをあらわす。すなわち、この条件が満たされたときのパラメータと流出高が最終的な解となる。

### (例題 13)

一価の非線形貯留方程式を用い、パラメータ  $k$  を 10、 $p$  を 0.7 として鶴川の出水事例の有効雨量 (表 1.2 参照) を与えて計算した結果を真値とし、初期値を適当に与えた場合、最適化により設定値に収束することをチェックせよ。なお、 $k$  の初期値は 50、計算の収束許容条件  $\varepsilon$  は 0.001 とする。

(例題 13 の解答)

具体的手順については先に示してある例題 9 を参照されたい。手順を簡単に箇条書きにまとめると次のようになる。

- [1] パラメータ値  $k$  のもと、式 (5.14) により、感度係数  $u$  を計算する。
- [2] パラメータ値  $k$  のもと、式 (5.8) により、流出高  $q$  を計算する。
- [3] 感度係数  $u$  を式 (5.12) により変数変換し、 $v$  を求める。
- [4] 実測値と計算値の差  $e$  および  $v$  から、式 (5.11) でパラメータ更新量を求める。
- [5] パラメータ  $k$  を更新し、式 (5.15) が満たされるまで [1]~[4] を繰り返す。

計算の考え方は以上のみだが、この他実際の計算上必要となるテクニックを述べる。まず、計算時間間隔だが、非線形問題の場合、流出高、感度係数を求めるためには実測データが入手される 1hr を 5 等分して 0.2hr 程度とすることが望ましい。また、計算過程でパラメータ  $k$  が負値をとらないように処理する(負値をとらないような初期値を設定する)とともに、次のような緩和係数  $fac$  を更新値  $\Delta k$  に乗じた方が収束が良くなるようである。

$$fac = 0.5(1 + fac)$$

$fac$  は当初 0 にセットしておき、すなわち第 1 回目の値が 0.5 となるようにして、以下繰り返しの過程で徐々に 1 に漸近するようにする。これは、初期の段階でパラメータ更新時に生じる発散を防御する処理といえる。

表 5.3 繰り返し計算によるパラメータの推移 ( $p = 0.7$ )

繰返回数	緩和係数 $fac$	更新値 $\Delta k$	更新前 $k$	打切条件 $ \Delta k/k $	更新後 $k$
1	0.50	-66.90	50.00	0.6690	16.55
2	0.75	-7.28	16.55	0.3298	11.09
3	0.88	-0.87	11.09	0.0686	10.33
4	0.94	-0.14	10.33	0.0126	10.20
5	0.97	-0.02	10.20	0.0018	10.18
6	0.98	0.00	10.18	0.0002	10.18




図 5.3 最適化による模擬流出データの再現結果 ( $p = 0.7$ )

以上の計算により、収束するまでのパラメータの推移を表 5.3 に示す。また、真値 (模擬データ) と最適化で得られた最終結果を表 5.4 および図 5.3 に示す。この結果収束時のパラメータは 10.18 となっている。

表 5.4 最適化による模擬流出データの再現結果 ( $p = 0.7$ )

時間 (hr)	有効雨量 (mm/h)	流出高 模擬データ (mm/h)	流出高 再現計算 (mm/h)	時間 (hr)	有効雨量 (mm/h)	流出高 模擬データ (mm/h)	流出高 再現計算 (mm/h)
0	0.000	0.000	0.000	26	0.000	1.073	1.245
1	0.776	0.025	0.025	27	0.000	0.922	1.075
2	1.878	0.143	0.140	28	0.000	0.798	0.935
3	1.640	0.268	0.264	29	0.000	0.694	0.817
4	0.626	0.300	0.297	30	0.000	0.608	0.718
5	0.119	0.281	0.280	31	0.000	0.535	0.634
6	1.909	0.442	0.440	32	0.000	0.473	0.562
7	9.251	1.591	1.576	33	0.000	0.420	0.500
8	5.783	2.239	2.234	34	0.000	0.374	0.447
9	3.430	2.415	2.432	35	0.000	0.335	0.401
10	10.064	3.768	3.797	36	0.000	0.301	0.361
11	18.589	6.800	6.851	37	0.000	0.271	0.326
12	14.421	8.449	8.579	38	0.000	0.245	0.296
13	10.909	8.904	9.130	39	0.000	0.222	0.269
14	20.467	11.627	11.938	40	0.000	0.202	0.245
15	7.836	10.483	10.902	41	0.000	0.184	0.223
16	2.773	8.455	8.926	42	0.000	0.168	0.205
17	0.507	6.524	6.994	43	0.000	0.154	0.188
18	0.000	5.048	5.488	44	0.000	0.142	0.173
19	0.000	3.979	4.379	45	0.000	0.130	0.159
20	0.031	3.192	3.549	46	0.000	0.120	0.147
21	0.000	2.591	2.909	47	0.000	0.111	0.136
22	0.000	2.130	2.410	48	0.000	0.103	0.126
23	0.000	1.770	2.018	49	0.000	0.095	0.117
24	0.000	1.485	1.704	50	0.000	0.088	0.109
25	0.000	1.257	1.451				

## 5-2 二価の非線形貯留方程式による流出解析

### 5-2-1 星モデルの概略

前節では一価の非線形貯留方程式を扱ってきたが、流出現象の再現精度をより向上させるために二価関数を用いるのは自然な方向である。ただし、数学的なモデルを構築することはいくらかでも可能だが、このレベルまでくると物理現象とのすり合わせをいかに図るかということも必要となってくる。なお、以後の解説について詳細を知りたい方は別途文献<sup>5)</sup>を参照されたい。

流出現象を物理モデルによって表現する方法として等価粗度法 (Kinematic wave 法) があるが、星らはその解との比較から  $s - q$  関係をあらわすのに次式が最適であるとしている。

$$s = k_1 q^{p_1} + k_2 \frac{d}{dt}(q^{p_2}) \quad (5.16)$$

$$\frac{ds}{dt} = r - q \quad (5.17)$$

ここで、 $k_1$ ,  $k_2$ ,  $p_1$ ,  $p_2$  がモデルパラメータであり、これらを等価粗度法のパラメータと比較し、

$$k_1 = 2.823(n/\sqrt{i})^{0.6} A^{0.24}, \quad k_2 = 0.2835 k_1^2 r^{-0.2648}, \quad p_1 = 0.6, \quad p_2 = 0.4648 \quad (5.18)$$

という結果が得られている。ここで、 $n$  はマンニングの粗度係数、 $i$  は斜面勾配、 $A$  は流域面積 ( $\text{km}^2$ )、 $\bar{r}$  は平均有効雨量強度 ( $\text{mm/h}$ ) をあらわす。特に上式中の  $(n/\sqrt{i})^{0.6}$  はフリクションファクターと称し、 $f_c$  で記述される。この  $f_c$  が流出の性格を特徴づけるもので、これが小さくなれば、すなわち粗度が小さくなるほど、もしくは勾配が急になるほどピーク流量が大きくなり、ピーク生起時間も早める作用をもつものである。しかしながら、現在  $f_c$  を一般的に求めうる理論を我々はもっていない。 $k_1$  および  $k_2$  はこの  $f_c$  に左右されるほか、 $k_2$  は有効雨量にも影響されるので、流出率についても物理的根拠によって求める必要があるが、これも至難なことである。そこで、モデルの物理的根拠を尊重するために、指数部のパラメータ  $p_1$  と  $p_2$  は式 (5.18) の値を用い、一方、流出現象の再現性を考えて  $k_1$  と  $k_2$  を数学的最適化手法によって探索することを今後の方針とする。

### 5-2-2 線形近似による解法

式 (5.16) および式 (5.17) を解いて流出高を求める方法を以下に示す。最初に次のような変数変換をおこなう。

$$y_1 = q^{p_2}, \quad y_2 = \frac{d}{dt}(q^{p_2}) \quad (5.19)$$

これから、式 (5.16) は次のようにあらわされる。

$$s = k_1 y_1^{p_1/p_2} + k_2 y_2 \quad (5.20)$$

式 (5.20) を式 (5.17) に代入すると、

$$\frac{dy_1}{dt} = y_2, \quad \frac{dy_2}{dt} = -\frac{k_1 p_1}{k_2 p_2} y_1^{p_1/p_2 - 1} y_2 - \frac{1}{k_2} y_1^{1/p_2} + \frac{1}{k_2} r \quad (5.21)$$

と整理できる。なお、この時の初期条件 ( $t = 0$ ) は  $y_1 = 0$ 、 $y_2 = 0$  と設定する。上式を  $Y = (y_1 \ y_2)^T$  によりベクトル表示すると、

$$\frac{d}{dt} \begin{pmatrix} y_1 \\ y_2 \end{pmatrix} = \frac{dY}{dt} = F(Y) \quad (5.22)$$

となる。なお、 $F(Y)$  の成分は式 (5.21) に示されている。これはスカラーの場合と同様に次のように線形化することができる。

$$\frac{dY}{dt} = AY + X \quad (5.23)$$

これによって、まさに線形二価貯留方程式を解いたときと同様に、次のような漸化式から  $Y$  の数値解が得られる。

$$Y_{m+1} = \Phi Y_m + \Gamma X_m \quad (5.24)$$

ここで、各ベクトル、マトリックスは次のようにあらわされる。

$$A = \begin{bmatrix} 0 & 1 \\ a_1 & a_2 \end{bmatrix}, \quad X_m = \begin{bmatrix} 0 \\ x_2 \end{bmatrix}, \quad \Phi = \begin{bmatrix} \phi_{11} & \phi_{12} \\ \phi_{21} & \phi_{22} \end{bmatrix}, \quad \Gamma = \begin{bmatrix} \gamma_{11} & \gamma_{12} \\ \gamma_{21} & \gamma_{22} \end{bmatrix} \quad (5.25)$$

この中の成分は、表 5.5 に示される。また、 $m$  は時間ステップをあらわす。

表 5.5 貯留方程式の数値解に関わる諸変数 (二価非線形)

$$\begin{aligned}
 a_1 &= -\frac{k_1}{k_2} \frac{p_1}{p_2} \left( \frac{p_1}{p_2} - 1 \right) y_{1m}^{p_1/p_2-2} y_{2m} - \frac{1}{k_2} \frac{1}{p_2} y_{1m}^{1/p_2-1} \\
 a_2 &= -\frac{k_1}{k_2} \frac{p_1}{p_2} y_{1m}^{p_1/p_2-1}, \quad a_3 = a_1 + a_2^2, \quad a_4 = a_1 + a_3 \\
 x_2 &= \frac{k_1}{k_2} \frac{p_1}{p_2} \left( \frac{p_1}{p_2} - 1 \right) y_{1m}^{p_1/p_2-1} y_{2m} + \frac{1}{k_2} \left( \frac{1}{p_2} - 1 \right) y_{1m}^{1/p_2} + \frac{1}{k_2} r_{m+1} \\
 d_1 &= -\frac{1}{k_2} \frac{p_1}{p_2} y_{1m}^{p_1/p_2-1} y_{2m}, \quad d_2 = \frac{1}{k_2^2} \left( k_1 \frac{p_1}{p_2} y_{1m}^{p_1/p_2-1} y_{2m} + y_{1m}^{1/p_2} - r_{m+1} \right) \\
 \phi_{11} &= 1 + \frac{1}{2} a_1 T^2 + \frac{1}{6} a_1 a_2 T^3 + \frac{1}{24} a_1 a_3 T^4 \\
 \phi_{12} &= T \left( 1 + \frac{1}{2} a_2 T + \frac{1}{6} a_3 T^2 + \frac{1}{24} a_2 a_4 T^3 \right) \\
 \phi_{21} &= a_1 T \left( 1 + \frac{1}{2} a_2 T + \frac{1}{6} a_3 T^2 + \frac{1}{24} a_2 a_4 T^3 \right) = a_1 \phi_{12} \\
 \phi_{22} &= 1 + a_2 T + \frac{1}{2} a_3 T^2 + \frac{1}{6} a_2 a_4 T^3 + \frac{1}{24} (a_1 a_3 + a_2^2 a_4) T^4 \\
 \gamma_{11} &= T \left( 1 + \frac{1}{6} a_1 T^2 + \frac{1}{24} a_1 a_2 T^3 \right) \\
 \gamma_{12} &= T^2 \left( \frac{1}{2} + \frac{1}{6} a_2 T + \frac{1}{24} a_3 T^2 \right) \\
 \gamma_{21} &= a_1 T^2 \left( \frac{1}{2} + \frac{1}{6} a_2 T + \frac{1}{24} a_3 T^2 \right) = a_1 \gamma_{12} \\
 \gamma_{22} &= T \left( 1 + \frac{1}{2} a_2 T + \frac{1}{6} a_3 T^2 + \frac{1}{24} a_2 a_4 T^3 \right) = \phi_{12}
 \end{aligned}$$

以上のマトリックス計算がある分計算はやや煩雑であるが、基本的な考え方はスカラーと同様に1階線形の常微分方程式を解くことにあることだけを念頭に置いてほしい。なお、得られた $Y$ から実際の流出高 $q$ を得るには、式(5.19)の変換を逆におこない、 $q = y_1^{1/p_2}$ とすればよい。

### 5-2-3 最適化手法によるパラメータの同定

二価の非線形貯留方程式の場合、2個のパラメータ $k_1$ と $k_2$ を最適化手法によって同定する必要がある。その更新量は次のマトリックス計算によって与えられる。

$$\Delta K = [V^T V]^{-1} [V^T E] \quad (5.26)$$

ここで,

$$\Delta K = \begin{bmatrix} \Delta k_1 \\ \Delta k_2 \end{bmatrix}, \quad V^T V = \begin{bmatrix} \sum v_{1i}^2 & \sum v_{1i} v_{2i} \\ \sum v_{1i} v_{2i} & \sum v_{2i}^2 \end{bmatrix}, \quad V^T E = \begin{bmatrix} \sum e_i v_{1i} \\ \sum e_i v_{2i} \end{bmatrix} \quad (5.27)$$

である。なお、 $N$ 個あるデータのうち  $i$  番目のデータについて考えると、 $e_i$  は計算流出高  $q_i$  と実測流出高  $q_i^*$  の差で、 $e_i = q_i^* - q_i$  である。また、 $V$  は感度係数行列といわれ、 $v_{1i} = \partial q_i / \partial k_1$ 、 $v_{2i} = \partial q_i / \partial k_2$  で定義される。この感度係数行列の成分を変数変換により、次のようにあらわす。式 (5.19) に示したような変数  $y_1, y_2$  に関する感度係数  $U$  は次のようにあらわされる。

$$U = [u_{11} \ u_{12} \ u_{21} \ u_{22}]^T = \begin{bmatrix} \frac{\partial y_1}{\partial k_1} & \frac{\partial y_1}{\partial k_2} & \frac{\partial y_2}{\partial k_1} & \frac{\partial y_2}{\partial k_2} \end{bmatrix}^T \quad (5.28)$$

である。この際の  $u$  と  $v$  の関係は次の通りである。

$$\begin{aligned} v_{1i} &= \frac{\partial q_i}{\partial k_1} = \frac{1}{p_2} y_i^{1/p_2-1} \frac{\partial y_{1i}}{\partial k_1} = \frac{1}{p_2} y_i^{1/p_2-1} u_{11i} \\ v_{2i} &= \frac{\partial q_i}{\partial k_2} = \frac{1}{p_2} y_i^{1/p_2-1} \frac{\partial y_{1i}}{\partial k_2} = \frac{1}{p_2} y_i^{1/p_2-1} u_{12i} \end{aligned} \quad (5.29)$$

ところで、 $U$  であらわされる感度係数行列は、貯留方程式の解法で示したのと同様に 1 階線形の常微分方程式であらわされることがわかっている。

$$\frac{dU}{dt} = A_1 U + D \quad (5.30)$$

また、その数値解は、

$$U_{m+1} = \Phi_1 U_m + \Gamma_1 D_m \quad (5.31)$$

となる。ここで、

$$\begin{aligned} A_1 &= \left[ \begin{array}{cc|cc} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ \hline a_1 & 0 & a_2 & 0 \\ 0 & a_1 & 0 & a_2 \end{array} \right], \quad D_m = \begin{bmatrix} 0 \\ 0 \\ d_1 \\ d_2 \end{bmatrix} \\ \Phi_1 &= \left[ \begin{array}{cc|cc} \phi_{11} & 0 & \phi_{12} & 0 \\ 0 & \phi_{11} & 0 & \phi_{12} \\ \hline \phi_{21} & 0 & \phi_{22} & 0 \\ 0 & \phi_{21} & 0 & \phi_{22} \end{array} \right], \quad \Gamma_1 = \left[ \begin{array}{cc|cc} \gamma_{11} & 0 & \gamma_{12} & 0 \\ 0 & \gamma_{11} & 0 & \gamma_{12} \\ \hline \gamma_{21} & 0 & \gamma_{22} & 0 \\ 0 & \gamma_{21} & 0 & \gamma_{22} \end{array} \right] \end{aligned} \quad (5.32)$$

であり、 $A_1$ 、 $\Phi_1$  および  $\Gamma_1$  の成分は先に示したもの (表 5.5 参照) と全く同様である。また、 $D_m$  の成分についても表 5.5 に示してある。結局、これも同様に式 (5.24) により  $Y$ 、式 (5.31) により感度係数  $U$  を求め、式 (5.26) によりパラメータ  $K$  を更新し、最適値が得られるまで計算を繰り返すわけである。なお、繰り返し計算の打ち切り条件は次の通りである。

$$\left| \frac{\Delta k_{1n}}{k_{1n}} \right| < \varepsilon, \quad \left| \frac{\Delta k_{2n}}{k_{2n}} \right| < \varepsilon \quad (5.33)$$

ここで、 $n$  は繰り返しステップをあらわす。ここまでできてわかったと思うが、微分方程式の階数が増えても同じアナロジーで解が得られるわけである。そこにはスカラー計算がマトリックス計算になる違いがあるだけである。すなわち線形近似法を適用することで、線形貯留方程式を解く際の基本ルーチンがそのまま利用できるという利点が生かされ続けるわけである。

#### (例題 14)

二価の非線形貯留方程式を用い、パラメータ  $k_1$  を 10、 $k_2$  を 10 とし、鶴川の出水事例の有効雨量 (表 1.2 参照) を与えて計算した結果を真値とし、初期値を適当に与えた場合、最適化により設定値に収束することをチェックせよ。なお、指数部のパラメータは  $p_1 = 0.6$ 、 $p_2 = 0.4648$  に固定して考えよ。

#### (例題 14 の解答)

具体的手順については先に示してある例題 10 を参照されたい。手順を簡単に箇条書きにまとめると次のようになる。

- [1] パラメータ値  $k_1$ 、 $k_2$  のもと、式 (5.31) により、感度係数行列の成分  $U$  を計算する。
- [2] パラメータ値  $k_1$ 、 $k_2$  のもと、式 (5.24) により、 $y_1$ 、 $y_2$  を計算する。
- [3] 感度係数行列  $U$  を式 (5.29) により変数変換し、 $V$  を求める。
- [4] 実測値と計算値の差  $e$  および  $V$  から、式 (5.26) でパラメータ更新量を求める。
- [5] パラメータ  $k_1$ 、 $k_2$  を更新し、式 (5.33) が満たされるまで [1]~[4] を繰り返す。

計算の流れは例題 13 とほとんど同じだが、本例の場合マトリックス計算があるので、計算ルーチンの構築はやや煩雑な作業となる。また、式 (5.33) を解く際に成分回帰分析手法を利用し<sup>2)</sup>、計算上の発散を防ぐなどの処置がなされている。

解答例では初期値として  $k_1$  を 20、 $k_2$  を 20、収束許容限界値  $\varepsilon$  を 0.001 とし、収束にいたるまでのパラメータの推移を表 5.6 に示す。また、真値 (模擬データ) と最適化で得られた最終結果を表 5.7 および図 5.4 に示す。この結果収束時のパラメータは設定値 ( $k_1 = 10$ 、 $k_2 = 10$ ) に一致している。

#### (例題 15)

二価の非線形貯留方程式を用い、鶴川の出水事例 (表 1.2 参照) を対象に最適化をおこなってパラメータを同定し、再現性をチェックせよ。なお、指数部のパラメータは  $p_1 = 0.6$ 、 $p_2 = 0.4648$  に固定して考えよ。

#### (例題 15 の解答)

考え方は例題 14 と同様で、計算プログラムも同じである。

解答例では初期値として  $k_1$  を 20、 $k_2$  を 20、収束許容限界値  $\varepsilon$  を 0.001 とし、収束にいたるまでのパラメータの推移を表 5.5 に示す。また、実測値と最適化で得られた最終結果を表 5.8 および図 5.5 に示す。この結果収束時のパラメータは  $k_1 = 21.71$ 、 $k_2 = 112.78$  となった。結果から線形貯留方程式を用いたものより再現精度の向上がみられている。なお、このような非線形貯留方程式を基本とした星モデルの再現性は、北海道内の様々な河川流量データを用いて検証されている。

表 5.6 繰り返し計算によるパラメータの推移 ( $p_1 = 0.6, p_2 = 0.4648$ )

繰返回数	更新値 $\Delta k_1$	更新前 $k_1$	打切条件 $ \Delta k_1/k_1 $	更新後 $k_1$	更新値 $\Delta k_2$	更新前 $k_2$	打切条件 $ \Delta k_2/k_2 $	更新後 $k_2$
1	-5.45	20.00	0.3745	14.55	-23.84	20.00	1.1919	20.00
2	-3.06	14.55	0.2658	11.50	-14.17	20.00	2.4320	5.83
3	-1.57	11.50	0.1582	9.93	2.63	5.83	0.3110	8.46
4	0.03	9.93	0.0028	9.95	1.19	8.46	0.1230	9.64
5	0.03	9.95	0.0035	9.99	0.30	9.64	0.0297	9.94
6	0.01	9.99	0.0010	10.00	0.05	9.94	0.0050	9.99
7	0.00	10.00	0.0002	10.00	0.01	9.99	0.0007	10.00




図 5.4 最適化による模擬流出データの再現結果 ( $p_1 = 0.6, p_2 = 0.4648$ )

表 5.7 最適化による模擬流出データの再現結果 ( $p_1 = 0.6, p_2 = 0.4648$ )

時間 (hr)	有効雨量 (mm/h)	流出高 模擬データ (mm/h)	流出高 再現計算 (mm/h)
0	0.000	0.000	0.000
1	0.776	0.001	0.001
2	1.878	0.020	0.019
3	1.640	0.097	0.096
4	0.626	0.196	0.195
5	0.119	0.249	0.248
6	1.909	0.332	0.331
7	9.251	0.937	0.936
8	5.783	2.100	2.099
9	3.430	2.874	2.874
10	10.064	4.057	4.057
11	18.589	7.312	7.313
12	14.421	10.804	10.805
13	10.909	11.942	11.943
14	20.467	14.026	14.027
15	7.836	13.710	13.710
16	2.773	10.452	10.451
17	0.507	7.183	7.181
18	0.000	4.864	4.863
19	0.000	3.402	3.401
20	0.031	2.488	2.488
21	0.000	1.891	1.891
22	0.000	1.481	1.481
23	0.000	1.189	1.189
24	0.000	0.974	0.974
25	0.000	0.811	0.811
26	0.000	0.685	0.684
27	0.000	0.585	0.585
28	0.000	0.505	0.505
29	0.000	0.440	0.440
30	0.000	0.386	0.386
31	0.000	0.341	0.341
32	0.000	0.304	0.304
33	0.000	0.272	0.272
34	0.000	0.244	0.244
35	0.000	0.221	0.221
36	0.000	0.200	0.200
37	0.000	0.182	0.182
38	0.000	0.167	0.167
39	0.000	0.153	0.153
40	0.000	0.141	0.141
41	0.000	0.130	0.130
42	0.000	0.120	0.120
43	0.000	0.111	0.111
44	0.000	0.103	0.103
45	0.000	0.096	0.096
46	0.000	0.090	0.090
47	0.000	0.084	0.084
48	0.000	0.079	0.079
49	0.000	0.074	0.074
50	0.000	0.069	0.069

表 5.8 繰り返し計算によるパラメータの推移 ( $p_1 = 0.6, p_2 = 0.4648$ )

繰返回数	更新値 $\Delta k_1$	更新前 $k_1$	打切条件 $ \Delta k_1/k_1 $	更新後 $k_1$	更新値 $\Delta k_2$	更新前 $k_2$	打切条件 $ \Delta k_2/k_2 $	更新後 $k_2$
1	3.44	20.00	0.1468	23.44	17.90	20.00	0.4724	37.90
2	1.17	23.44	0.0477	24.62	40.54	37.90	0.5168	78.44
3	-2.31	24.62	0.1033	22.31	29.65	78.44	0.2743	108.09
4	-0.57	22.31	0.0264	21.74	4.01	108.09	0.0358	112.10
5	-0.03	21.74	0.0012	21.71	0.60	112.10	0.0053	112.70
6	0.00	21.71	0.0000	21.71	0.08	112.70	0.0007	112.77




図 5.5 最適化による実測流出データの再現結果 ( $p_1 = 0.6, p_2 = 0.4648$ )

表 5.9 最適化による実測流出データの再現結果 ( $p_1 = 0.6, p_2 = 0.4648$ )

時間 (hr)	有効雨量 (mm/h)	流出高 実測 (mm/h)	流出高 再現計算 (mm/h)
0	0.000	0.000	0.000
1	0.776	0.001	0.000
2	1.878	0.002	0.000
3	1.640	0.011	0.001
4	0.626	0.021	0.005
5	0.119	0.030	0.010
6	1.909	0.049	0.018
7	9.251	0.104	0.044
8	5.783	0.217	0.113
9	3.430	0.402	0.231
10	10.064	0.652	0.429
11	18.589	1.130	0.840
12	14.421	2.032	1.583
13	10.909	3.095	2.624
14	20.467	4.232	4.045
15	7.836	6.172	5.712
16	2.773	7.311	7.099
17	0.507	7.678	7.931
18	0.000	7.877	8.172
19	0.000	7.754	7.942
20	0.031	7.458	7.399
21	0.000	6.944	6.684
22	0.000	6.287	5.905
23	0.000	5.560	5.135
24	0.000	4.573	4.419
25	0.000	3.800	3.778
26	0.000	3.169	3.219
27	0.000	2.714	2.740
28	0.000	2.407	2.335
29	0.000	2.115	1.994
30	0.000	1.939	1.710
31	0.000	1.736	1.472
32	0.000	1.557	1.274
33	0.000	1.385	1.109
34	0.000	1.179	0.970
35	0.000	1.050	0.854
36	0.000	0.926	0.755
37	0.000	0.857	0.672
38	0.000	0.777	0.601
39	0.000	0.699	0.541
40	0.000	0.646	0.489
41	0.000	0.583	0.444
42	0.000	0.521	0.405
43	0.000	0.472	0.371
44	0.000	0.423	0.341
45	0.000	0.376	0.315
46	0.000	0.349	0.291
47	0.000	0.303	0.271
48	0.000	0.268	0.253
49	0.000	0.242	0.236
50	0.000	0.208	0.221

## あとがき

これまで、線形から非線形にいたる貯留関数法を解説してきた。最終的には非線形二価の貯留関数、いわゆる星モデルの解法まで到達したが、これが現在北海道開発局の洪水予測システムの中で全部局を網羅している流出モデルである。星モデルが完全に普及している事実は当研究所のアンケート調査で明らかになったわけだが、かつて大学の研究室レベルにあったノウハウを、これまで現場に普及させるためには長い期間を要したことになる。同時に、このような優れた手法とそれを実現する強力なコンピュータ、自動化されたシステムは時代の流れとしても、それらを過信することにも危惧を覚える。実際の洪水の最中にはオンラインシステムは往々にして止まるものである。また、時として自然現象は複雑な様相を呈し、我々がそのすべてを理解しているわけでもない。つまり、モデル化できていない要因も多々あるのである。例えば、抵抗則や損失雨量(流出率)の物理的解釈とモデル化は、未だ解決をみない大きな問題である。これがあるゆえ機械的に演算される結果を鵜呑みにすると、実態と合わないことに慌てふためくこともあり得る。さらに言うと、システム自体がブラックボックスで、何に手をつければよいかわからないという弊害も生ずる。ただし、そこでお手上げでは河川管理者としての責任は免れないであろう。いざとなれば、水文情報をオフライン処理したり、予測精度を少しでも向上させるようにシステムの問題点(モデルパラメータ等)を是正していく作業が必要となる。そのためにも、これまで解説してきた方法論を役立てていただきたいのである。

また、最近北海道開発局でもレーダー雨量計のネットワークを完備しつつあり、これでもたらされる雨量情報の洪水予測システムへの活用を考え、当研究所でも調査研究を進めている。時代の進歩とともに新しい技術の活用、また流出モデル自体の改良に対しても現場レベルで問題意識をもっていただければ次の進歩が約束されるであろう。

これまで連載してきた「現場のための水文学」はここで一応締めとする。これは数年前(土木試験所時代)刊行した「現場のための水理学」との兄弟版とも言えるものである。当研究所河川部門の主要テーマである水理、水文でこのようなテキストを揃えたことで、担当者としては1つの責任が果たせた気がする。なお、補遺に掲載されたプログラムについては、請求があればフロッピーディスクで随時提供していきたい。これも著作にあたった者の重要な責任と考える。

(中津川 誠)

## 補遺 計算プログラムの概要

### 例題 12

非線形一価貯留方程式について、線形近似法とルンゲクッタギル法の比較

#### (1) プログラムの変数

DT:            サンプル時間間隔  
 TN:            解析の全時間  
 P, KK:        モデルパラメータ  
 ALF, BETA:    パラメータの変換値  
 MB:           時間分割数  
 TT:           計算時間ステップ  
 T:            実測データ入手時間  
 Y:            線形近似法の流出高  
 YP:           流出高変換値  
 Q:            Runge-Kutta-Gill 法の流出高  
 R:            有効雨量  
 SS:           ルンゲクッタギル法の変数  
 AA:           微分方程式の係数  $a$   
 BB:           微分方程式の係数  $x$   
 F:            微分方程式の解の係数  $\phi$   
 G:            微分方程式の解の係数  $\gamma$

#### (2) プログラムの解説

文番号	解説
1040~1100	パラメータ等の設定
1120~1170	データの読み込み
1200~1320	線形近似法による演算
1340~1490	Runge-Kutta-Gill 法による演算
1550~1880	Runge-Kutta-Gill 法サブルーチン
1900~2420	ディスプレイ画面表示サブルーチン

#### (3) プログラムリスト その1

```

1000 '   save "c:hikaku.bas",a
1010 REM 非線形一価関数（漸化式法とルンゲクッタギル法の比較）
1020 REM S=Ky^p
1030 CLS 3
1040 DT=1   :' サンプル時間間隔
1050 TN=50  :' 解析の全時間
1060 KK=20  :' パラメータ
1070 P=.7   :' パラメータ
1080 MB=5   :' 時間分割数
1090 TT=DT/MB
1100 ALF=-1/KK : BETA=1/P
1110 DIM T(200),Y(200),Q(200),R(200),SS(200)
1120 OPEN "c:mukawa00.out" FOR INPUT AS #1
1130 OPEN "c:hikaku.out" FOR OUTPUT AS #2
1140 WHILE NOT EOF(1)
1150 INPUT #1,I,DUMMY,R(I)
1160 T(I)=I
1170 WEND
1180 '
1190 REM 漸化式法
1200 Y(I)=0 : YP=0
1210 FOR I=1 TO TN
1220 FOR J=1 TO MB
1230 AA=ALF*BETA*YP^(BETA-1)
1240 BB=ALF*(1-BETA)*YP^BETA-ALF*R(I)
1250 AT=AA*TT
1260 F=1+AT+1/2*(AT)^2+1/6*(AT)^3+1/24*(AT)^4
1270 G=(1+1/2*AT+1/6*(AT)^2+1/24*(AT)^3)*TT
1280 YP=YP*F+BB*G
1290 IF YP<0 THEN YP=0
1300 NEXT J
1310 Y(I)=YP^BETA
1320 NEXT I
1330 '
    
```

## 補遺 計算プログラムの概要

### (3) プログラムリスト その2

```
1340 REM ルンゲクッタギル法
1350 Q(0)=0
1360 SU=TN      :'上限値(x)
1370 SL=T(0)   :'下限値(x)
1380 SD=DT     :'刻み幅
1390 SS(0)=Q(0) :'初期値(y)
1400 NI=MB     :'サンプル点間の分割数
1410 L=INT((SU-SL)/SD)+1
1420 GOSUB *GILL
1430 I=0
1440 X=SL
1450 Q(I)=SS(I)^BETA
1460 PRINT #2,T(I),Y(I),Q(I),R(I)
1470 X=X+SD
1480 I=I+1
1490 IF X<SU THEN 1450
1500 '
1510 GOSUB *PLOT
1520 CLOSE:END
1530 '
1540 'ルンゲクッタギル法のサブルーチン
1550 *GILL
1560 S1=.2928932188134528#
1570 S2=.1213203435596426#
1580 S3=.5857864376269054#
1590 S4=1.707106781186548#
1600 S5=-4.121320343559643#
1610 S6=3.414213562373097#
1620 SX=SL
1630 SY=SS(0)
1640 SZ=SL
1650 I=1
1660 DEF FN SF(X,Y)=ALF*Y^BETA-ALF*R(I)
1670 SH=SD/NI
1680 FOR K=1 TO NI
1690  SB=SH*FN SF(SX,SY)
1700  SY=SY+.5*SB
1710  SC=SB
1720  SB=SH*FN SF(SX+.5*SH,SY)
1730  SY=SY+S1*(SB-SC)
1740  SC=S2*SC+S3*SB
1750  SB=SH*FN SF(SX+.5*SH,SY)
1760  SY=SY+S4*(SB-SC)
1770  SC=S5*SC+S6*SB
1780  SB=SH*FN SF(SX+SH,SY)
1790  SY=SY+SB/6-SC/3
1800  SX=SX+SH
1810 NEXT K
1820 SS(I)=SY
1830 SZ=SZ+SD
1840 SX=SZ
1850 IF SZ>=SU THEN RETURN
1860 I=I+1
1870 GOTO 1660
1880 RETURN
1890 '
```

## 補遺 計算プログラムの概要

### (3) プログラムリスト その3

```

1900 '作図用サブルーチン
1910 *PLOT
1920 CLS
1930 XMAX=-10000:XMIN=10000:YMAX=-10000:YMIN=10000
1940 FOR I=0 TO TN
1950 IF T(I)>XMAX THEN XMAX=T(I)
1960 IF T(I)<XMIN THEN XMIN=T(I)
1970 IF Q(I)>YMAX THEN YMAX=Q(I)
1980 IF Q(I)<YMIN THEN YMIN=Q(I)
1990 IF Y(I)>YMAX THEN YMAX=Y(I)
2000 IF Y(I)<YMIN THEN YMIN=Y(I)
2010 NEXT I
2020 'PRINT USING "XMAX=#####.## XMIN=#####.##";XMAX,XMIN
2030 'PRINT USING "YMAX=#####.## YMIN=#####.##";YMAX,YMIN
2040 '
2050 'INPUT " X の最大値";XMA
2060 'INPUT " X の最小値";XMI
2070 'INPUT " X の目盛り間隔";XK
2080 'INPUT " Y の最大値";YMA
2090 'INPUT " Y の最小値";YMI
2100 'INPUT " Y の目盛り間隔";YK
2110 'XMA=3:XMI=1:XK=.25
2120 XMA=50 :XMI=0 :XK=5
2130 YMA=20 :YMI=0 :YK=5
2140 CLS
2150 '
2160 SCREEN 3,0,0,1
2170 VIEW(0,0)-(399,399):WINDOW(0,0)-(399,399)
2180 XS=350/(XMA-XMI):YS=350/(YMA-YMI)
2190 X0=10:Y0=10
2200 X1=X0:Y1=Y0:X2=(XMA-XMI)*XS+X0:Y2=(YMA-YMI)*YS+Y0
2210 LINE(X1,Y1)-(X2,Y2),7,B
2220 FOR XX=XMI TO XMA STEP XK
2230 X1=(XX-XMI)*XS+X0:Y1=(YMA-YMI)*YS+Y0
2240 X2=X1:Y2=Y0
2250 LINE(X1,Y1)-(X2,Y2),7
2260 NEXT XX
2270 FOR YY=YMI TO YMA STEP YK
2280 X1=X0:Y1=(YMA-YY)*YS+Y0
2290 X2=(XMA-XMI)*XS+X0:Y2=Y1
2300 LINE(X1,Y1)-(X2,Y2),7
2310 NEXT YY
2320 FOR I=1 TO TN
2330 X20=(T(I)-XMI)*XS+X0 : Y20=(YMA-Q(I))*YS+Y0
2340 X11=(T(I-1)-XMI)*XS+X0: Y11=(YMA-Y(I-1))*YS+Y0
2350 X21=(T(I)-XMI)*XS+X0 : Y21=(YMA-Y(I))*YS+Y0
2360 CIRCLE(X20,Y20),3,1,,,F
2370 LINE(X11,Y11)-(X21,Y21),7
2380 NEXT I
2390 'INPUT "これをコピーしますか(y/n)";YN2$
2400 'IF YN2$="y" OR YN2$="Y" THEN CLS:COPY
2410 'CLS 3
2420 RETURN


```

## 補遺 計算プログラムの概要

### 例題 13

非線形一価貯留方程式について、最適化によるパラメータ同定

#### (1) プログラムのフローチャート



## 補遺 計算プログラムの概要

### (2) プログラムの変数

DT:	サンプリング時間間隔
TN:	解析の全時間
EPSILON:	収束許容限界値
P, K:	モデルパラメータ
MB:	時間分割数
TT:	計算時間ステップ
T:	実測データ入手時間
Y:	流出高計算値
YP:	流出高変換値
Q:	流出高模擬値
R:	有効雨量
UP:	感度係数
AA:	微分方程式の係数 $a$
BB:	微分方程式の係数 $x$
F:	微分方程式の解の係数 $\phi$
G:	微分方程式の解の係数 $\gamma$
XX:	感度係数の二乗値
XE:	感度係数と誤差の積
EE:	実測値と計算値の誤差
SEE:	誤差二乗和
FAC:	パラメータ更新時の緩和係数
DK:	パラメータ更新値

### (3) プログラムの解説

文番号	解説
1050~1110	パラメータ等の設定
1140~1220	データの読み込み
1340~1510	感度係数, 流出高の逐次計算
1550~1610	パラメータの更新

ディスプレイ画面表示サブルーチンは例題  
12のプログラムリストを参照。

### (4) プログラムリスト その1

```

1000 ' save "c:run11-1.bas",a
1010 REM 貯留関数に非線形一価関数をあてはめた場合(漸化式法)
1020 REM 最適化によるパラメータの決定(pは固定)
1030 REM S=Ky^pの形での流出解析
1040 CLS 3
1050 TN=50          :'解析の全時間
1060 DT=1          :'時間間隔
1070 EPSILON=.001  :'打ち切り誤差
1080 K=50          :'パラメータ初期値
1090 P=.7          :'パラメータ初期値
1100 MB=5          :'時間分割数
1110 TT=DT/MB
1120 DIM T(200),Y(200),Q(200),R(200)
1130 '
1140 'データ読み込み
1150 OPEN "c:mukawa01.out" FOR INPUT AS #1
1160 OPEN "c:mukawa2.out" FOR OUTPUT AS #2
1170 INPUT #1,I,Q(I),R(I)
1180 T(I)=I
1190 WHILE NOT EOF(1)
1200 INPUT #1,I,Q(I),R(I)
1210 T(I)=I
1220 WEND
1230 Y(0)=0
1240 '
1250 TIMES="00:00:00"
    
```

## 補遺 計算プログラムの概要

### (4) プログラムリスト その2

```

1260 '*****
1270 '* 流出高の推定と最適化 *
1280 '*****
1290 YP=0
1300 UP=0
1310 XX=0
1320 XE=0
1330 SEE=0
1340 FOR I=1 TO TN
1350   FOR J=1 TO MB
1360     AA=-1/K/P*YP^(1/P-1)
1370     BB=-1/K*(1-1/P)*YP^(1/P)+1/K*R(I)
1380     AT=AA*TT
1390     F=1+AT+1/2*AT^2+1/6*AT^3+1/24*AT^4
1400     G=(1+1/2*AT+1/6*AT^2+1/24*AT^3)*TT
1410     UP=UP*F+(YP^(1/P)-R(I))/K/K*G
1420     YP=YP*F+BB*G
1430     IF YP<0 THEN YP=0
1440   NEXT J
1450   IF YP>0 THEN Y(I)=YP^(1/P) ELSE Y(I)=0
1460   IF YP>0 THEN CC=1/P*YP^(1/P-1) ELSE CC=0
1470   EE=Q(I)-Y(I) : '誤差
1480   XX=XX+UP*UP*CC^2
1490   XE=XE+UP*EE*CC
1500   SEE=SEE+EE*EE
1510 NEXT I
1520 '
1530 SEE=SQR(SEE/TN)
1540 'パラメータ更新
1550 FAC=.5*(1+FAC)
1560 DK=XE/XX
1570 IF ABS(DK/K)<EPSILON THEN 1710
1580 DK=DK*FAC
1590 XNEW=K+DK
1600 IF XNEW>0 THEN K=K+DK
1610 COUNT=COUNT+1
1620 CLS
1630 '
1640 GOSUB *PLOT
1650 LOCATE 50,1:PRINT "繰り返し回数=";COUNT
1660 LOCATE 50,2:PRINT "パラメータ K=";K
1670 LOCATE 50,3:PRINT "パラメータ p=";P
1680 LOCATE 50,4:PRINT "経過時間    =" ;TIMES
1690 GOTO 1270
1700 '
1710 CLS
1720 GOSUB *PLOT
1730 FOR I=0 TO TN
1740 'PRINT I,Y(I),Q(I)
1750 PRINT #2,I,Y(I),Q(I),R(I)
1760 NEXT I
1770 LOCATE 50,1:PRINT "繰り返し回数=";COUNT
1780 LOCATE 50,2:PRINT "パラメータ K=";K
1790 LOCATE 50,3:PRINT "パラメータ p=";P
1800 LOCATE 50,4:PRINT "誤差 2 乗和    =" ;SEE
1810 LOCATE 50,5:PRINT "経過時間    =" ;TIMES
1820 CLOSE : END


```

## 補遺 計算プログラムの概要

### 例題 14 および 15

非線形二価貯留方程式について、最適化によるパラメータ同定

#### (1) プログラムのフローチャート



## (2) プログラムの変数

DT:	サンプリング時間間隔
TN:	解析の全時間
N:	微分方程式の階数
NP:	パラメータの個数
EPSILON:	収束許容限界値
K1, K2:	モデルパラメータ (係数部)
P1, P2:	モデルパラメータ (指数部)
AI:	モデルパラメータ (係数部)
MB:	時間分割数
TT:	計算時間ステップ
T:	実測データ入手時間
Y:	流出高計算値
YI:	流出高変換値
YP:	前ステップの流出高変換値
Q:	流出高実測値
R:	有効雨量
U, UU:	感度係数
A1~A4:	行列 A の成分 $a_1 \sim a_4$
B2:	行列 X の成分 $x_2$
C:	行列 D の成分 $d_1 \sim d_2$
F:	微分方程式の解の係数 $\phi$
G:	微分方程式の解の係数 $\gamma$
EE:	実測値と計算値の誤差
SEE:	誤差二乗和
XX:	感度係数の二乗成分行列
W:	行列 XX の対角成分
PP:	規準化行列の成分
LL:	コレスキー (下三角) 行列の成分
LU:	上三角行列の成分
UI:	上三角行列の逆行列成分
DAI:	パラメータ更新値
FAC:	パラメータ更新時の緩和係数

## (3) プログラムの解説

文番号	解説
1120~1240	パラメータ等の設定
1270~1320	データの読み込み
1380~1780	感度係数, 流出高の逐次計算 サブルーチン*MATCAL : $\Phi, \Gamma$ 算出 サブルーチン*CALU : 感度係数の $U$ 算出 サブルーチン*CALY : 流出高変換の $Y$ 算出
1810~1830	感度係数の二乗値 $X^T X$ の算出 サブルーチン*CALXX : 感度係数の二乗値 $X^T X$ の算出
1870~1950	成分回帰分析手法による パラメータ更新 サブルーチン*CALP : 規準化行列 PP の算出 サブルーチン*CALLL : コレスキー行列 LL の算出 サブルーチン*CALLT : 上三角行列 LU の算出 サブルーチン*CALG : 上三角行列の逆行列 UI の算出 サブルーチン*CALA : パラメータ更新値 DAI の算出
2000~2120	データ出力

ディスプレイ画面表示サブルーチンは例題 12 のプログラムリストを参照。

## 補遺 計算プログラムの概要

### (4) プログラムリスト その1

```

1000 '   save "c:run22-1.bas",a
1010 REM  貯留関数に非線形二価関数(星モデル)をあてはめた場合(漸化式法)
1020 REM  最適化によるパラメータの決定
1030 REM  成分回帰分析手法の適用
1040 REM  S=K1 y`p1+K2 dy`p2/dtの形での流出解析
1050 CLS 3
1060 DIM T(200),Y(200),Q(200),R(200),YI(10),YP(10),C(10)
1070 DIM F(10,10),G(10)
1080 DIM U(20),UU(20),PP(10,10),LL(10,10),LU(10,10),W(10),AI(10),DAI(10)
1090 DIM X(200,10),XX(10,10),ZZ(200,10),GG(200,10),COE(20)
1100 OPEN "c:mukawa1.dat" FOR INPUT AS #1
1110 OPEN "c:mukawa4.out" FOR OUTPUT AS #2
1120 TN=50 : '解析の全時間
1130 DT=1  : '解析時間間隔(hr)
1140 MB=5  : '時間分割数
1150 N=2   : '微分方程式の階数
1160 NP=2  : 'パラメータの個数
1170 TT=DT/MB
1180 NP1=NP+1
1190 NNP=N*NP
1200 EPSILON=.001 : '打ち切り誤差
1210 ILP=100
1220 K1=20      : K2=20      : 'パラメータ初期値
1230 P1=.6      : P2=.4648   : '指数部パラメータ初期値
1240 AI(1)=K1  : AI(2)=K2
1250 LOCATE 30,10 : PRINT "計算実施中"
1260 '
1270 ' 原データ(有効雨量)の読み込み
1280 '  Q(I) 実測直接流出高(mm/h), R(I) 有効雨量(mm/h)
1290 WHILE NOT EOF(1)
1300   INPUT #1,I,Q(I),R(I) :T(I)=I
1310   IF I=>TN THEN 1330
1320 WEND
1330 '
1340 ' *****
1350 '*   最適化による繰り返し計算   *
1360 ' *****
1370 '
1380 FOR I=1 TO N : YP(I)=0 : NEXT I
1390 FOR I=1 TO NNP : U(I)=0 : NEXT I
1400 K1=AI(1) : K2=AI(2) : K0=K1/K2 : P0=P1/P2
1410 '
1420 SEE=0
1430 FOR II=1 TO TN
1440   FOR L=1 TO MB
1450     '
1460     ' パラメータの設定
1470     IF YP(1)=<0 THEN GOTO 1520
1480     AA1=YP(1)^(P0-2)
1490     AA2=YP(1)^(1/P2-1)
1500     AA3=YP(1)^(P0-1)
1510     AA4=YP(1)^(1/P2)
1520     A1=-K0*P0*(P0-1)*AA1*YP(2)-1/K2/P2*AA2
1530     A2=-K0*P0*AA3
1540     A3=A1+A2*AA2
1550     A4=A1+A3
1560     B2=K0*P0*(P0-1)*YP(1)^(P0-1)*YP(2)+1/K2*(1/P2-1)*YP(1)^(1/P2)+R(II)/K2
1570     C(1)=-1/K2*P0*AA3*YP(2)
1580     C(2)=(K1*P0*AA3*YP(2)+AA4-R(II))/K2/K2
1590     '

```

## 補遺 計算プログラムの概要

### (4) プログラムリスト その2

```

1600 '漸化式マトリックス(F,G)の計算
1610 ' F(I,J) マトリックスΦの成分, G(I) マトリックスΓの成分
1620 GOSUB *MATCAL
1630 '
1640 '漸化式による最適化(パラメータの決定)と流出高,感度係数の計算
1650 ' Y(II) 計算直接流出高(mm/h), YI(I) 変数変換後の成分
1660 GOSUB *CALU
1670 GOSUB *CALY
1680 IF YP(1)<0 THEN YP(1)=0
1690 IF YP(1)>0 THEN CC=1/P2*YP(1)^(1/P2-1) ELSE CC=0
1700 FOR I=1 TO NP : X(II,I)=U(I)*CC : NEXT I : '感度係数行列
1710 NEXT L
1720 IF YI(1)>0 THEN Y(II)=YI(1)^(1/P2) ELSE Y(II)=0
1730 EE=Q(II)-Y(II)
1740 X(II,NP1)=EE
1750 IF Y(II)<0 THEN Y(II)=0
1760 SEE=SEE+EE*EE
1770 '
1780 NEXT II
1790 SEE=SQR(SEE/TN) : '誤差2乗和
1800 '
1810 'マトリックス XXの算出
1820 ' EE(I) 誤差, XX(I,J) マトリックスXX
1830 GOSUB *CALXX
1840 '
1850 GOSUB *PLOT : '作図
1860 '
1870 '最適化
1880 '成分回帰分析手法によるパラメータ更新量の算定
1890 GOSUB *CALP : '規準化行列の算出
1900 GOSUB *CALLL : 'コレスキー行列(下三角行列)の算出
1910 GOSUB *CALLT : '上三角行列の算出
1920 GOSUB *CALG : '上三角行列の逆行列算出
1930 GOSUB *CALA : 'パラメータ更新量ベクトルの算出
1940 IF ILP<1 THEN 1990 : '繰り返し条件
1950 COUNT=COUNT+1 : '繰り返し回数
1960 GOTO 1350
1970 '
1980 '*****
1990 '
2000 FOR II=0 TO TN
2010 'PRINT II,Y(II),Q(II)
2020 PRINT #2,II,Y(II),Q(II)
2030 NEXT II
2040 CLS
2050 LOCATE 50,1 :PRINT "収束回数=";COUNT
2060 FOR I=1 TO N
2070 LOCATE 50,I+1 :PRINT "K";I;"=";AI(I)
2080 NEXT I
2090 LOCATE 50,N+3 :PRINT "誤差2乗和=";SEE
2100 LOCATE 50,20 :INPUT "これをコピーしますか(y/n)";YN2S
2110 IF YN2S="y" OR YN2S="Y" THEN COPY
2120 CLS 3
2130 CLOSE : END
2140 '

```

## 補遺 計算プログラムの概要

### (4) プログラムリスト その3

```
2150 ' サブルーチン
2160 '
2170 *MATCAL
2180 ' マトリックス計算のサブルーチン
2190 ' Set initial values of Matrices, F(I,J) and G(I)
2200 F(1,1)=1+1/2*A1*TT^2+1/6*A1*A2*TT^3+1/24*A1*A3*TT^4
2210 F(1,2)=TT*(1+1/2*A2*TT+1/6*A3*TT^2+1/24*A2*A4*TT^3)
2220 F(2,1)=A1*F(1,2)
2230 F(2,2)=1+A2*TT+1/2*A3*TT^2+1/6*A2*A4*TT^3+1/24*(A1*A3+A2*A2*A4)*TT^4
2240 G(1)=TT^2*(1/2+1/6*A2*TT+1/24*A3*TT^2)
2250 G(2)=F(1,2)
2260 RETURN
2270 '
2280 *CALY
2290 ' 流出高の算出
2300 FOR I=1 TO N
2310 YI(I)=0
2320   FOR J=1 TO N
2330     YI(I)=YI(I)+F(I,J)*YP(J)
2340   NEXT J
2350   YI(I)=YI(I)+G(I)*B2
2360 NEXT I
2370 FOR I=1 TO N : YP(I)=YI(I) : NEXT I
2380 RETURN
2390 '
2400 *CALU
2410 ' 感度係数マトリックスXの算出
2420 I1=0
2430 FOR I=1 TO N
2440   FOR K=1 TO NP
2450     I1=I1+1
2460     S=0
2470     FOR J=1 TO N
2480       J1=K+NP*(J-1)
2490       S=S+F(I,J)*U(J1)
2500     NEXT J
2510     S=S+G(I)*C(K)
2520     UU(I1)=S
2530   NEXT K
2540 NEXT I
2550 FOR I=1 TO I1
2560   U(I)=UU(I)
2570 NEXT I
2580 RETURN
2590 '
2600 *CALXX
2610 ' マトリックスXXの計算
2620 FOR I=1 TO N
2630   FOR J=1 TO I
2640     S=0
2650     FOR K=1 TO TN
2660       S=S+X(K,I)*X(K,J) : ' マトリックス XX
2670     NEXT K
2680     XX(I,J)=S
2690   NEXT J
2700 NEXT I
2710 RETURN
2720 '
```

## 補遺 計算プログラムの概要

### (4) プログラムリスト その4

```
2730 ' 規準化行列の算出
2740 ' PP(I,J) 規準化行列
2750 *CALP
2760 FOR I=1 TO N : W(I)=SQR(XX(I,I)) : NEXT I
2770 FOR I=1 TO N
2780   S=W(I)
2790   FOR J=1 TO I
2800     S1=W(J)
2810     XX(I,J)=XX(I,J)/(S*S1)
2820     XX(J,I)=XX(I,J)
2830   NEXT J
2840 NEXT I
2850 FOR I=1 TO N:FOR J=1 TO N:PP(I,J)=XX(I,J):NEXT J:NEXT I
2860 RETURN
2870 '
2880 ' コレスキー分解
2890 ' LL(I,J) コレスキー行列 (下三角行列)
2900 *CALLL
2910 FOR J=1 TO N-1
2920   LL(J,J)=ABS(PP(J,J))
2930   LL(J,J)=SQR(LL(J,J))
2940   AL=1/LL(J,J)
2950   FOR K=N TO J+1 STEP -1
2960     LL(K,J)=AL*PP(K,J)
2970     BE=LL(K,J)
2980     FOR I=K TO N
2990       PP(I,K)=PP(I,K)-LL(I,J)*BE
3000     NEXT I
3010   NEXT K
3020 NEXT J
3030 LL(N,N)=ABS(PP(N,N))
3040 LL(N,N)=SQR(LL(N,N))
3050 RETURN
3060 '
3070 ' 上三角行列の算出
3080 ' LU(I,J) 上三角行列
3090 *CALLT
3100 FOR I=1 TO N
3110   FOR J=I TO N
3120     LU(I,J)=LL(J,I)
3130   NEXT J
3140 NEXT I
3150 RETURN
3160 '
3170 ' 上三角行列の逆行列算出
3180 ' UI(I,J) 上三角行列の逆行列
3190 *CALG
3200 UI(1,1)=1/LU(1,1)
3210 FOR J=2 TO N
3220   UI(J,J)=1/LU(J,J)
3230   JM=J-1
3240   FOR K=1 TO JM
3250     S=0
3260     FOR I=K TO JM
3270       S=S-UI(K,I)*LU(I,J)
3280     NEXT I
3290   UI(K,J)=S*UI(J,J)
3300 NEXT K
3310 NEXT J
3320 RETURN
3330 '
```

## 補遺 計算プログラムの概要

### (4) プログラムリスト その5

```
3340 'パラメータ更新量ベクトルの算出
3350 ' DAI(I) パラメータ更新量ベクトル
3360 *CALA
3370 FOR J=1 TO N
3380 S=W(J)
3390 FOR I=1 TO TN
3400 ZZ(I,J)=X(I,J)/S
3410 NEXT I
3420 NEXT J
3430 FOR I=1 TO TN
3440 FOR J=1 TO N
3450 S=0
3460 FOR K=1 TO J
3470 S=S+ZZ(I,K)*UI(K,J)
3480 NEXT K
3490 GG(I,J)=S
3500 NEXT J
3510 NEXT I
3520 FOR I=1 TO N
3530 S=0
3540 FOR J=1 TO TN
3550 S=S+GG(J,I)*X(J,NP1)
3560 NEXT J
3570 COE(I)=S
3580 NEXT I
3590 FOR I=1 TO N
3600 S=0
3610 FOR J=I TO N
3620 S=S+UI(I,J)*COE(J)
3630 NEXT J
3640 DAI(I)=S/W(I)
3650 NEXT I
3660 FAC=.5*(1+FAC)
3670 FOR I=1 TO N :DAI(I)=FAC*DAI(I) : NEXT I
3680 '
3690 ILP=0
3700 LOCATE 50,1 :PRINT "繰り返し回数=";COUNT
3710 FOR I=1 TO N
3720 ANEW=AI(I)+DAI(I)
3730 IF ANEW>0 THEN AI(I)=AI(I)+DAI(I)
3740 DD=ABS(DAI(I)/AI(I))
3750 LOCATE 50,I+1 :PRINT "打切り基準値=";DD
3760 IF DD>EPSILON THEN ILP=ILP+1
3770 NEXT I
3780 LOCATE 50,N+1 :PRINT " 誤差二乗和=";SEE
3790 LOCATE 50,N+2 :PRINT "  パラメータ K1=";AI(1)
3800 LOCATE 50,N+3 :PRINT "  パラメータ K2=";AI(2)
3810 RETURN
```

## 補遺 だれにでもできる星モデルのパラメータ同定

ここでは、詳しい理論の理解は抜きに、BASIC プログラムを起動してパラメータの最適同定をおこなう手順を解説する。これまで、開土研月報の「やさしい…」とか「よくわかる…」というタイトルにだまされた人も多数いると思うが、この部分は本当にタイトル通りなので安心されたい。パソコンさえ動かせれば、数式アレルギーの人でもこの部分だけは見てもらえると確信する。なお、プログラムリストは例題 15 の補遺を参照していただきたい。この際、内容の理解は省略してよい。まずは Try あるのみ!

- 1 図 A のようなデータセット (左から時間 (hr), 直接流出高 (mm/h), 有効雨量 (mm/h)) を用意せよ。生の流量データから直接流出高, 有効雨量を求める方法は参考文献 1) を参照せよ。データファイルは MS-DOS のテキスト形式で作っておくと汎用性がある。例題 15 ではデータファイル名を mukawa1.dat として、フロッピーディスクに save している。(この部分だけは前処理としてややテクニックが必要だがデータの処理自体は小学校の算数ができれば可能である。)
- 2 パソコンの電源を入れ、BASIC を起動せよ。
- 3 データファイルとプログラムの入ったフロッピーディスク (FD) をディスクセットに挿入せよ。この時のドライブ名を確認しておくこと。例題 15 では FD のドライブは c である。
- 4 BASIC ファイルの run22-1.bas を呼び出せ。この時、load "c:run22-1.bas" とタイプする。
- 5 結果を入出力するファイル名を指定せよ。例題 15 では 1100 行で入力ファイルとして、mukawa1.dat, 1110 行で出力ファイルとして、mukawa4.out を設定している。
- 6 解析の時間関係のパラメータを設定せよ。例題 15 では 1120 行で総時間数 50 時間, 1130 行で時間間隔 1 時間, 1140 行で 1 時間を何分割するかを設定している。この時、1 時間を 5 分割して 0.2 時間間隔で実際の計算を実施している。
- 7 モデルパラメータの初期値を設定せよ。例題 15 では 1220 行で  $k_1$  を 20,  $k_2$  を 20 としている。指数部のパラメータは固定して考えるので、いじる必要はないが、もし線形モデルを考えたいならば、1230 行で  $p_1$  を 1,  $p_2$  を 1 とすればよい。
- 8 以上の設定が終わった後、run とタイプせよ。プログラムが起動し、自動的にパラメータの最適解を計算してくれる。このとき、画面のディスプレイ上には図 B のように実測値との適合度が逐次出力される。
- 9 データファイルは mukawa4.out に出力されているので、その中をみて数値をチェックすること

以上

時 間	実測流出高	有効雨量
0	0.000	0.000
1	0.001	0.776
2	0.002	1.878
3	0.011	1.640
4	0.021	0.626
5	0.030	0.119

}

図 A データファイルの構造




図 B ディスプレイ出力例

なお、わからないことがあったら開発土木研究所に相談のこと。